

CALCUL FORMEL ET MÉCANIQUE DU POINT MATÉRIEL CHARGÉ

Alain Dupont

lycée Jean Perrin
Lyon 9°

On se propose ici de montrer sur quelques exemples simples, issus de séances de travaux dirigés avec des élèves de MPSI, comment l'emploi du calcul formel permet de résoudre rapidement et surtout systématiquement quelques exercices classiques de mécanique du point matériel chargé.

Le logiciel ici utilisé est **MAPLE™** dont dispose le lycée.

les cas traités sont les suivants:

- A) mouvement dans un champ magnétostatique uniforme $\vec{B}$ sans frottement .
- B) la situation A) avec un frottement visqueux.
- C) la situation A) avec en plus un champ électrostatique uniforme $\vec{E}$ orthogonal à $\vec{B}$.

Le référentiel d'étude sera considéré galiléen .

La charge q de la particule est supposée positive;

la particule sera toujours non relativiste et son poids sera négligé devant les forces liées à l'existence de la charge électrique.

Toutes ces situations se traitent de la même manière:

- création du vecteur position de la particule

- création de son vecteur vitesse et de son vecteur accélération

- création du vecteur force résultante

- écriture de la relation fondamentale de la dynamique de Newton

- projection de cette relation sur les 3 axes du repère, conduisant au système différentiel à intégrer.

- intégration formelle du système, compte tenu des conditions initiales, conduisant aux lois horaires du mouvement

- représentation de la trajectoire

Les exemples ont été choisis de sorte que l'intégration formelle du système différentiel soit possible.

Ce n'est bien sûr pas le cas général et il faut alors se contenter d'une intégration numérique qui permet le tracé de la trajectoire sans expliciter les lois horaires.

On notera la nécessité de simplifier les expressions souvent très lourdes fournies par le logiciel : ceci reste un point noir du calcul formel.

La méthode illustrée par ces exemples est facilement transposable à d'autres problèmes de mécanique du point (dans le champ de pesanteur, dans un champ de force central coulombien...)

On présente ici 3 feuilles de calcul correspondant successivement aux 3 cas étudiés.

Toutes les courbes représentatives sont groupées en fin d'article.

Bibliographie:

Fortin, Pomès, *Premiers pas en Maple*, ed. Vuibert

Leroux, Pomès, *Toutes les applications de Maple*, ed. Vuibert

A.Heck, *Introduction to Maple*, ed. Springer Verlag

EXEMPLE A

- particule chargée dans un champ magnétostatique B (seul agissant dans un référentiel galiléen)
- $B=B_z = \text{Cste}$
- position initiale : l'origine du repère ; vitesse initiale : v_0 contenue dans le plan xOz
- on recherche les lois horaires $x(t), y(t), z(t)$
- on représente ensuite la trajectoire.

> restart;

> assume(q>0,B>0,m>0);

> with(linalg,vector,crossprod);

[crossprod, vector]

on charge les outils vecteur et produit vectoriel, dont on a besoin ensuite

> OM:=vector([x(t),y(t),z(t)]);#vecteur position

$$OM := [x(t) \quad y(t) \quad z(t)]$$

> v:=map(u->diff(u,t),OM);#vecteur vitesse

$$v := \left[\frac{\partial}{\partial t} x(t) \quad \frac{\partial}{\partial t} y(t) \quad \frac{\partial}{\partial t} z(t) \right]$$

noter l'instruction map qui distribue la fonction précisée sur les éléments d'un tableau

> acc:=map(u->diff(u,t),v);#vecteur accélération

$$acc := \left[\frac{\partial^2}{\partial t^2} x(t) \quad \frac{\partial^2}{\partial t^2} y(t) \quad \frac{\partial^2}{\partial t^2} z(t) \right]$$

> champB:=vector([0,0,B]);

$$champB := [0 \quad 0 \quad B]$$

> Lorentz:=map(u->q*u,crossprod(v,champB));#force de Lorentz

$$Lorentz := \left[q \left(\frac{\partial}{\partial t} y(t) \right) B \quad -q \left(\frac{\partial}{\partial t} x(t) \right) B \quad 0 \right]$$

> mgamma:=map(u->u*m,acc);#dérivée/t de la quantité de mouvement

$$mgamma := \left[\left(\frac{\partial^2}{\partial t^2} x(t) \right) m \quad \left(\frac{\partial^2}{\partial t^2} y(t) \right) m \quad \left(\frac{\partial^2}{\partial t^2} z(t) \right) m \right]$$

> evalm(Lorentz)=evalm(mgamma);#RFD appliquée à la particule

$$\left[q \left(\frac{\partial}{\partial t} y(t) \right) B \quad -q \left(\frac{\partial}{\partial t} x(t) \right) B \quad 0 \right] = \left[\left(\frac{\partial^2}{\partial t^2} x(t) \right) m \quad \left(\frac{\partial^2}{\partial t^2} y(t) \right) m \quad \left(\frac{\partial^2}{\partial t^2} z(t) \right) m \right]$$

on projette sur les trois axes Ox,Oy,Oz

> eqx:=mgamma[1]=Lorentz[1];

$$eqx := \left(\frac{\partial^2}{\partial t^2} x(t) \right) m = q \left(\frac{\partial}{\partial t} y(t) \right) B$$

> eqy:=mgamma[2]=Lorentz[2];

$$eqy := \left(\frac{\partial^2}{\partial t^2} y(t) \right) m = -q \left(\frac{\partial}{\partial t} x(t) \right) B$$

> eqz:=mgamma[3]=Lorentz[3];

$$eqz := \left(\frac{\partial^2}{\partial t^2} z(t) \right) m = 0$$

on obtient ainsi le système différentiel auquel conduit la RFD

on résoud d'abord eq2

> dsolve({eqz,z(0)=0,D(z)(0)=voz},z(t));

$$z(t) = voz \, t$$

noter l'emploi de dsolve avec des conditions initiales

> assign("");#affectation de la solution à z(t)

le mouvement projeté sur z'z est uniforme

> sys:=eqx,eqy;

$$sys := \left(\frac{\partial^2}{\partial t^2} x(t) \right) m \sim = q \sim \left(\frac{\partial}{\partial t} y(t) \right) B \sim, \left(\frac{\partial^2}{\partial t^2} y(t) \right) m \sim = -q \sim \left(\frac{\partial}{\partial t} x(t) \right) B \sim$$

> dsolve({sys,x(0)=0,y(0)=0,D(x)(0)=vox,D(y)(0)=0},{x(t),y(t)},method=laplace);

$$\left\{ y(t) = -\frac{m \sim vox}{q \sim B \sim} + \frac{m \sim vox \cos\left(\frac{q \sim B \sim t}{m \sim}\right)}{q \sim B \sim}, x(t) = \frac{m \sim vox \sin\left(\frac{q \sim B \sim t}{m \sim}\right)}{q \sim B \sim} \right\}$$

NB) si on ne précise pas la méthode de résolution, la solution présentée par Maple est très lourde

(l'auteur conseille de faire le test)

> assign("");

on dispose ici des lois horaires x(t) et y(t) qu'on va simplifier.

on introduit la pulsation cyclotron omega=q*B/m

> eq4:=m*w=q*B;

$$eq4 := m \sim w = q \sim B \sim$$

> x:=factor(simplify(x(t),{eq4},{B}));#on elimine B des équations en tenant comp

> te de eq4

$$x := \frac{vox \sin(t \, w)}{w}$$

> y:=factor(simplify(y(t),{eq4},{B}));

$$y := \frac{vox (-1 + \cos(t \, w))}{w}$$

pour une représentation graphique de la trajectoire on introduit des valeurs numériques à savoir ici: w=10E6rad/s, vox=3*10E5m/s=voz

> zz:=subs(w=1E6,voz=3E5,z(t));

$$zz := 300000. \, t$$

> xx:=subs(w=1E6,vox=3E5,x);

$$xx := .3 \sin(.1 \, 10^7 \, t)$$

> yy:=subs(w=1E6,vox=3E5,y);

$$yy := -.3 + .3 \cos(.1 \, 10^7 \, t)$$

> with(plots,spacecurve);

$$[spacecurve]$$


> spacecurve([xx,yy,zz],t=0..20E-6,numpoints=400,color=black,axes=boxed,title

> ='partie A, particule chargée dans B uniforme');

on obtient la trajectoire hélicoïdale attendue dans les conditions initiales choisies.

>

partie A, particule chargée dans B uniforme



EXEMPLE B

- trajectoire d'une particule chargée dans un champ magnétostatique B avec frottement visqueux, c'est à dire proportionnel à la vitesse ; $f = -k.v$

- (ici on notera lambda le coefficient de frottement par unité de masse : ie $\lambda = k / m$)

- $B = B_z = \text{Cste}$

- position initiale: $x_0 = a > 0, y_0 = 0, z_0 = 0$

- vecteur vitesse initiale v_0 contenu dans le plan xOy : $v_{0x} = -a.\lambda, v_{0y} = a.w$
(ce choix du vecteur vitesse initial permet une simplification des calculs)

- on recherche les lois horaires $x(t), y(t), z(t)$

NB : pour les instructions déjà rencontrées, on a ici inhibé l'impression du résultat (en les terminant par : au lieu de ;)

```
> restart;

> assume(q>0,B>0,m>0,a>0,k>0,lambda>0,w>0);

> with(linalg,vector,crossprod):

> OM:=vector([x(t),y(t),z(t)]):#vecteur position

> v:=map(u->diff(u,t),OM):#vecteur vitesse

> acc:=map(u->diff(u,t),v):#vecteur accélération

> champB:=vector([0,0,-B]):

> Lorentz:=map(u->q*u,crossprod(v,champB)):#force de Lorentz

> mgamma:=map(u->u*m,acc):

> frot:=map(u->-k*u,v):

> F:=evalm(Lorentz+frot):#résultante

> evalm(F)=evalm(mgamma):#RFD appliquée à la particule
```

$$\left[-q \sim \left(\frac{\partial}{\partial t} y(t) \right) B \sim -k \sim \left(\frac{\partial}{\partial t} x(t) \right) \quad q \sim \left(\frac{\partial}{\partial t} x(t) \right) B \sim -k \sim \left(\frac{\partial}{\partial t} y(t) \right) \quad -k \sim \left(\frac{\partial}{\partial t} z(t) \right) \right] =$$

$$\left[\left(\frac{\partial^2}{\partial t^2} x(t) \right) m \sim \left(\frac{\partial^2}{\partial t^2} y(t) \right) m \sim \left(\frac{\partial^2}{\partial t^2} z(t) \right) m \sim \right]$$

on projette sur les trois axes

> eqx:=mgamma[1]=F[1];

$$eqx := \left(\frac{\partial^2}{\partial t^2} x(t) \right) m \sim -q \sim \left(\frac{\partial}{\partial t} y(t) \right) B \sim -k \sim \left(\frac{\partial}{\partial t} x(t) \right)$$

> eqy:=mgamma[2]=F[2];

$$eqy := \left(\frac{\partial^2}{\partial t^2} y(t) \right) m \sim = q \sim \left(\frac{\partial}{\partial t} x(t) \right) B \sim -k \sim \left(\frac{\partial}{\partial t} y(t) \right)$$

> eqz:=mgamma[3]=F[3];

$$eqz := \left(\frac{\partial^2}{\partial t^2} z(t) \right) m \sim = -k \sim \left(\frac{\partial}{\partial t} z(t) \right)$$

on obtient ainsi le système différentiel auquel conduit la RFD

on résoud d'abord eqz

> dsolve({eqz,z(0)=0,D(z)(0)=0},z(t));

$$z(t) = 0$$

> assign("");

le mouvement est donc plan:dans xOy

> sys:=eqx,eqy;

$$\begin{aligned} sys := & \left(\frac{\partial^2}{\partial t^2} x(t) \right) m \sim = -q \sim \left(\frac{\partial}{\partial t} y(t) \right) B \sim -k \sim \left(\frac{\partial}{\partial t} x(t) \right), \\ & \left(\frac{\partial^2}{\partial t^2} y(t) \right) m \sim = q \sim \left(\frac{\partial}{\partial t} x(t) \right) B \sim -k \sim \left(\frac{\partial}{\partial t} y(t) \right) \end{aligned}$$

> dsolve({sys,x(0)=a,y(0)=0,D(x)(0)=-a*lambda,D(y)(0)=a*w},{x(t),y(t)},method=la
> place);

$$\begin{aligned} y(t) = & -\frac{a \sim m \sim q \sim B \sim \lambda \sim}{q \sim^2 B \sim^2 + k \sim^2} + \frac{a \sim m \sim k \sim w \sim}{q \sim^2 B \sim^2 + k \sim^2} + 2 \frac{a \sim m \sim^3 k \sim \lambda \sim \%2 \sin\left(\frac{q \sim B \sim t}{m \sim}\right)}{\%1} \\ & - \frac{a \sim m \sim^3 k \sim^2 w \sim \%2 \sin\left(\frac{q \sim B \sim t}{m \sim}\right)}{\%1 q \sim B \sim} + \frac{a \sim m \sim^3 w \sim q \sim B \sim \%2 \sin\left(\frac{q \sim B \sim t}{m \sim}\right)}{\%1} \\ & - \frac{a \sim m \sim^5 q \sim^2 B \sim^2 \lambda \sim \%2 \sin\left(\frac{q \sim B \sim t}{m \sim}\right) k \sim}{\%1^2} - \frac{a \sim m \sim^5 \lambda \sim \%2 \sin\left(\frac{q \sim B \sim t}{m \sim}\right) k \sim^3}{\%1^2} \\ & + \frac{a \sim m \sim^3 q \sim B \sim \lambda \sim \%2 \cos\left(\frac{q \sim B \sim t}{m \sim}\right)}{\%1} + \frac{a \sim m \sim^5 k \sim^2 w \sim \%2 \sin\left(\frac{q \sim B \sim t}{m \sim}\right) q \sim B \sim}{\%1^2} \\ & + \frac{a \sim m \sim^5 k \sim^4 w \sim \%2 \sin\left(\frac{q \sim B \sim t}{m \sim}\right)}{\%1^2 q \sim B \sim} - \frac{a \sim m \sim^3 k \sim w \sim \%2 \cos\left(\frac{q \sim B \sim t}{m \sim}\right)}{\%1}, x(t) = \\ & \frac{a \sim k \sim^2}{q \sim^2 B \sim^2 + k \sim^2} + \frac{a \sim q \sim^2 B \sim^2}{q \sim^2 B \sim^2 + k \sim^2} - \frac{a \sim m \sim w \sim q \sim B \sim}{q \sim^2 B \sim^2 + k \sim^2} - \frac{a \sim m \sim k \sim \lambda \sim}{q \sim^2 B \sim^2 + k \sim^2} \end{aligned}$$

$$\begin{aligned}
& + 2 \frac{a \sim m \sim^3 k \sim w \sim \%2 \sin\left(\frac{q \sim B \sim t}{m \sim}\right)}{\%1} + \frac{a \sim m \sim^3 k \sim^2 \lambda \sim \%2 \sin\left(\frac{q \sim B \sim t}{m \sim}\right)}{\%1 q \sim B \sim} \\
& - \frac{a \sim m \sim^3 \lambda \sim q \sim B \sim \%2 \sin\left(\frac{q \sim B \sim t}{m \sim}\right)}{\%1} - \frac{a \sim m \sim^5 w \sim q \sim^2 B \sim^2 \%2 \sin\left(\frac{q \sim B \sim t}{m \sim}\right) k \sim}{\%1^2} \\
& - \frac{a \sim m \sim^5 w \sim \%2 \sin\left(\frac{q \sim B \sim t}{m \sim}\right) k \sim^3}{\%1^2} + \frac{a \sim m \sim^3 w \sim q \sim B \sim \%2 \cos\left(\frac{q \sim B \sim t}{m \sim}\right)}{\%1} \\
& - \frac{a \sim m \sim^5 k \sim^2 \lambda \sim \%2 \sin\left(\frac{q \sim B \sim t}{m \sim}\right) q \sim B \sim}{\%1^2} - \frac{a \sim m \sim^5 k \sim^4 \lambda \sim \%2 \sin\left(\frac{q \sim B \sim t}{m \sim}\right)}{\%1^2 q \sim B \sim} \\
& + \frac{a \sim m \sim^3 k \sim \lambda \sim \%2 \cos\left(\frac{q \sim B \sim t}{m \sim}\right)}{\%1} \Bigg\} \\
& \%1 := q \sim^2 B \sim^2 m \sim^2 + k \sim^2 m \sim^2 \\
& \%2 := e^{\left(\frac{(-q \sim^2 B \sim^2 m \sim k \sim - k \sim^3 m \sim) t}{\%1} \right)}
\end{aligned}$$

on notera ici l'extrême lourdeur ,voire l'illisibilité,de la solution présentée.

> assign("");

on dispose ici des lois horaires x(t) et y(t) qu'il faut absolument simplifier

> eq4:=m*w=q*B;

$$eq4 := m \sim w \sim = q \sim B \sim$$

> eq5:=k=lambda*m;

$$eq5 := k \sim = m \sim \lambda \sim$$

> xx:=factor(simplify(x(t),{eq4,eq5},{q,k}));

$$xx := \frac{a \sim \cos(t \sim w \sim)}{e^{(t \sim \lambda \sim)}}$$

> yy:=factor(simplify(y(t),{eq4,eq5},{q,k}));

$$yy := \frac{a \sim \sin(t \sim w \sim)}{e^{(t \sim \lambda \sim)}}$$

on introduit ici des valeurs numériques pour le tracé de la trajectoire

> absc:=subs(a=5,w=10,lambda=0.3,xx);

> ord:=subs(a=5,w=10,lambda=0.3,yy);

> plot([absc,ord,t=0..10],title=`partie B,trajectoire de la particule freinée`);

>

la trajectoire est une spirale logarithmique comme on peut en voir sur les clichés de chambres à bulles.


On peut visualiser le renforcement du freinage sur l'aspect de la trajectoire en augmentant la valeur numérique de λ : ici 0.8 au lieu de 0.5

```
> absc:=subs(a=5,w=10,lambda=0.8,xx):
```


```
> ord:=subs(a=5,w=10,lambda=0.8,yy):
```

```
> plot([absc,ord,t=0..10],title=`partie B,trajectoire de la particule davantage frein  
> ée`);
```

partie B,trajectoire de la particule freinée



partie B, trajectoire de la particule davantage freinée



EXEMPLE C

- trajectoire d'une particule chargée dans deux champs croisés E et B
- $B = B_y = \text{Cste}$
- $E = E_x = E_0$
- position initiale: l'origine du repère
- vitesse initiale v_0 , colinéaire à Oz
- on recherche les lois horaires
- pour les instructions déjà rencontrées, on a ici encore inhibé l'impression du résultat

```

> restart;

> assume(q>0,B>0,m>0,Eo>0,vo>0);

> with(linalg,vector,crossprod):

> OM:=vector([x(t),y(t),z(t)]):

> v:=map(u->diff(u,t),OM):#vecteur vitesse

> acc:=map(u->diff(u,t),v):#vecteur accélération

> champB:=vector([0,B,0]):

> champE:=vector([Eo,0,0]):

> Lorentz:=map(u->q*u,crossprod(v,champB)):#force de Lorentz

> F:=evalm(Lorentz+map(u->q*u,champE)):#force résultante

> mgamma:=map(u->u*m,acc):

> evalm(F)=evalm(mgamma);#RFD

```

$$\begin{bmatrix} -q \sim \left(\frac{\partial}{\partial t} z(t) \right) B \sim + q \sim E_0 \sim & 0 & q \sim \left(\frac{\partial}{\partial t} x(t) \right) B \sim \\ \left[\left(\frac{\partial^2}{\partial t^2} x(t) \right) m \sim & \left(\frac{\partial^2}{\partial t^2} y(t) \right) m \sim & \left(\frac{\partial^2}{\partial t^2} z(t) \right) m \sim \end{bmatrix}$$

on projette sur les trois axes

```
> eqx:=mgamma[1]=F[1];
```

$$eqx := \left(\frac{\partial^2}{\partial t^2} x(t) \right) m \sim = -q \sim \left(\frac{\partial}{\partial t} z(t) \right) B \sim + q \sim E_0 \sim$$

> eqy:=mgamma[2]=F[2];

$$eqy := \left(\frac{\partial^2}{\partial t^2} y(t) \right) m \sim = 0$$

> eqz:=mgamma[3]=F[3];

$$eqz := \left(\frac{\partial^2}{\partial t^2} z(t) \right) m \sim = q \sim \left(\frac{\partial}{\partial t} x(t) \right) B \sim$$

on obtient ainsi le système différentiel auquel conduit la RFD
on résoud d'abord eqy

> dsolve({eqy,y(0)=0,D(y)(0)=0},y(t));

$$y(t) = 0$$

qui peut le plus peut le moins! Le mouvement est donc plan, dans xOz.

> sys:=eqx,eqz;

$$sys := \left(\frac{\partial^2}{\partial t^2} x(t) \right) m \sim = -q \sim \left(\frac{\partial}{\partial t} z(t) \right) B \sim + q \sim E o \sim, \left(\frac{\partial^2}{\partial t^2} z(t) \right) m \sim = q \sim \left(\frac{\partial}{\partial t} x(t) \right) B \sim$$

> dsolve({sys,x(0)=0,z(0)=0,D(x)(0)=0,D(z)(0)=vo},{x(t),z(t)},method=laplace);

$$\left\{ \begin{aligned} x(t) &= -\frac{m \sim v o \sim}{B \sim q \sim} + \frac{m \sim E o \sim}{B \sim^2 q \sim} + \frac{m \sim \cos\left(\frac{q \sim B \sim t}{m \sim}\right) v o \sim}{q \sim B \sim} - \frac{m \sim \cos\left(\frac{q \sim B \sim t}{m \sim}\right) E o \sim}{q \sim B \sim^2}, \\ z(t) &= \frac{t E o \sim}{B \sim} + \frac{m \sim \sin\left(\frac{q \sim B \sim t}{m \sim}\right) v o \sim}{B \sim q \sim} - \frac{m \sim \sin\left(\frac{q \sim B \sim t}{m \sim}\right) E o \sim}{B \sim^2 q \sim} \end{aligned} \right\}$$

> assign("");

on dispose ici des lois horaires x(t) et z(t)

> xx:=x(t);

$$xx := -\frac{m \sim v o \sim}{B \sim q \sim} + \frac{m \sim E o \sim}{B \sim^2 q \sim} + \frac{m \sim \cos\left(\frac{q \sim B \sim t}{m \sim}\right) v o \sim}{q \sim B \sim} - \frac{m \sim \cos\left(\frac{q \sim B \sim t}{m \sim}\right) E o \sim}{q \sim B \sim^2}$$

> zz:=z(t);

$$zz := \frac{t E o \sim}{B \sim} + \frac{m \sim \sin\left(\frac{q \sim B \sim t}{m \sim}\right) v o \sim}{B \sim q \sim} - \frac{m \sim \sin\left(\frac{q \sim B \sim t}{m \sim}\right) E o \sim}{B \sim^2 q \sim}$$

on introduit la pulsation cyclotron omega=q*B/m

> eq4:=m*w=q*B;

$$eq4 := m \sim w = q \sim B \sim$$

> x:=factor(simplify(xx,{eq4},{q}));#on elimine q des équations en tenant compte

> de eq4

> z:=factor(simplify(zz,{eq4},{q}));#idem

$$x := \frac{(B \sim v_0 - E_0) (-1 + \cos(t \, w))}{B \sim w}$$

$$z := \frac{t E_0 \sim w + \sin(t \, w) v_0 \sim B - \sin(t \, w) E_0}{B \sim w}$$

pour les représentations graphiques on introduit des valeurs numériques
par ex. un proton soumis à B=1mT et à un champ Eo=10V/m (soit Eo/B=10000m/s)
(w=96385.5 Hz)

a) vo=Eo/B

> eq5:=Eo=vo*B;

$$eq5 := E_0 = B \sim v_0$$

> x1:=simplify(x,{eq5},{Eo});

$$x1 := 0$$

> z1:=simplify(z,{eq5},{Eo});

$$z1 := v_0 \sim t$$

le mouvement est rectiligne uniforme selon Oz (les champs sont ignorés)
b) vo>Eo/B

> x2:=subs(B=1E-3,Eo=10,vo=0.5E5,w=96385.5,x);

$$x2 := -.4150001816 + .4150001816 \cos(96385.5 \, t)$$

> z2:=subs(B=1E-3,Eo=10,vo=0.5E5,w=96385.5,z);

$$z2 := 10000.00000 \, t + .4150001816 \sin(96385.5 \, t)$$

> plot([x2,z2,t=0..3E-4],title='partie C,vo>Eo/B');

c) vo<Eo/B

> x3:=subs(B=1E-3,Eo=10,vo=100,w=96385.5,x);

$$x3 := .1027125449 - .1027125449 \cos(96385.5 \, t)$$

> z3:=subs(B=1E-3,Eo=10,vo=100,w=96385.5,z);

$$z3 := 10000.00000 \, t - .1027125449 \sin(96385.5 \, t)$$

> plot([x3,z3,t=0..3E-4],title='partie C,vo<Eo/B');

>

partie C, $v_0 > E_0/B$ 